Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS
Fluid Dynamics



NOTICE

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].
   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.
   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].
   d. The printer indication of the multiply sign is ⁑. When you see ⁑ in the program listing, press [×].
   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).
   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■, followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL^{T}SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02^{T}THIS IS A*
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	*03^T ⊢ SAMPLE*
04 AVIEW	■ AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↗*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] ■ [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	*11^{T}R3=*
12 ARCL 03	■ [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	■ [RTN]	*14 RTN*

FLUID DYNAMICS AND HYDRAULICS

* Requires at least one additional memory module.

CONDUIT FLOW

This program solves for the average velocity, or the pressure drop for viscous, incompressible flow in conduits.

Equations:

$$v^2 = \frac{\Delta P/\rho}{2\left(f\,\frac{L}{D} + \frac{K_T}{4}\right)}$$

For laminar flow (Re < 2300)

$$f = 16/\text{Re}.$$

For turbulent flow (Re > 2300)

$$\frac{1}{\sqrt{f}} = 1.737\ \ln\frac{D}{\varepsilon} + 2.28 - 1.737\ \ln\left(4.67\,\frac{D}{\varepsilon \text{Re}\sqrt{f}} + 1\right)$$

is solved by Newton's method.

$$\frac{1}{\sqrt{f_0}} = 1.737\ \ln\frac{D}{\varepsilon} + 2.28$$

is used an an initial guess in the iteration.

where:
- Re is the Reynolds number, defined as $\rho D v/\mu$;
- D is the pipe diameter;
- ε is the dimension of irregularities in the conduit surface (see table 2);
- f is the Fanning friction factor for conduit flow;
- ΔP is the pressure drop along the conduit;
- ρ is the density of the fluid;
- μ is the viscosity of the fluid;
- ν is the kinematic viscosity of the fluid and $\mu=\rho\nu$;
- L is the conduit length;
- v is the average fluid velocity;
- K_T is the total of the applicable fitting coefficients in table 1.

Table 1

Fitting Coefficients

Fitting	K
Globe valve, wide open	7.5——10
Angle valve, wide open	3.8
Gate valve, wide open	0.15——0.19
Gate valve, 3/4 open	0.85
Gate valve, 1/2 open	4.4
Gate valve, 1/4 open	20
90° elbow	0.4——0.9
Standard 45° elbow	0.35——0.42
Tee, through side outlet	1.5
Tee, straight through	.4
180° bend	1.6
Entrance to circular pipe	0.25——0.50
Sudden expansion	$(1-A_{up}/A_{dn})^2$*
Acceleration from v=0 to $v=v_{entrance}$	1.0

*A_{up} is the upstream area and A_{dn} is the downstream area.

Table 2

Surface Irregularities

Material	ε(feet)	ε(meters)
Drawn or Smooth Tubing	$5.0x10^{-6}$	$1.5x10^{-6}$
Commercial Steel or Wrought Iron	$1.5x10^{-4}$	$4.6x10^{-5}$
Asphalted Cast Iron	$4.0x10^{-4}$	$1.2x10^{-4}$
Galvanized Iron	$5.0x10^{-4}$	$1.5x10^{-4}$
Cast Iron	$8.3x10^{-4}$	$2.5x10^{-4}$
Wood Stave	$6.0x10^{-4}$ to $3.0x10^{-3}$	$1.8x10^{-4}$ to $9.1x10^{-4}$
Concrete	$1.0x10^{-3}$ to $1.0x10^{-2}$	$3.0x10^{-4}$ to $3.0x10^{-3}$
Riveted Steel	$3.0x10^{-3}$ to $3.0x10^{-2}$	$9.1x10^{-4}$ to $9.1x10^{-3}$

Reference:

Welty, Wicks, Wilson, *Fundamentals of Momentum, Heat and Mass Transfer*, John Wiley and Sons, Inc., 1969.

Remarks:

The correlation gives meaningless results in the region 2300 < Re < 4000.

The solution requires an iterative procedure. The time for solution will range from 10 seconds for ΔP, to several minutes for v. The display setting is used to determine when the solution for v is adequately accurate. Time for solution of v is roughly proportional to the number or significant digits in the display setting.

If the conduit is not circular, an equivalent diameter may be calculated using the formula below:

$$D_{eq} = 4\frac{\text{cross sectional area}}{\text{wetted perimeter}}$$

Unitary consistency must be maintained.

Example:

A heat exchanger has 20, 3 meter tube passes (60 m of pipe) with 180 degrees bends connecting each pair of tubes (from table 1, $K_T = 10 \times 1.6$). The fluid is water ($\nu = 9.3 \times 10^{-7} m^2/s$, $\rho = 10^3 kg/m^3$). The surface roughness is 3×10^{-4}m and the diameter is 2.54×10^{-2}m. If the fluid velocity is 3.05 m/s, what is the pressure loss? What is the Reynolds number? What is the Fanning friction factor?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 015	
[///] [ENG] 3	
[XEQ] [ALPHA] CONDUIT [ALPHA]	U=?
9.3 [EEX] [CHS] 7 [ENTER↑]	
[EEX] 3 [X] [R/S]	RHO=?
[EEX] 3 [R/S]	E=?
3 [EEX] [CHS] 4 [R/S]	L=?
60 [R/S]	D=?
2.54 [EEX] [CHS] 2 [R/S]	KT=?
16 [R/S]	V=?
3.05 [R/S]	DP=?
[R/S]	DP=521.9E3
[R/S]	Re=83.30E3
[R/S]	F=10.18E-3

User Instructions

SIZE: 015

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize		[XEQ] CONDUIT	U=?
3	Key in viscosity	μ	[R/S]	RHO=?
4	Key in density	ρ	[R/S]	E=?
5	Key in surface irregularity	ε	[R/S]	L=?
6	Key in length of conduit	L	[R/S]	D=?
7	Key in equivalent diameter	D	[R/S]	KT=?
8	Key in total fitting coefficient	K_T	[R/S]	V=?
9	Key in velocity if known, otherwise			
	press [R/S]	v	[R/S]	DP=?
10	Key in pressure drop if known, otherwise			
	press [R/S]. (either v or ΔP must be			
	known)	ΔP	[R/S]	(unknown)=
11	Optional: obtain Reynolds no.		[R/S]	Re=
	Optional: obtain Fanning friction factor		[R/S]	F=
12	To change just v or ΔP		[XEQ] C	V=?
13	Go to step 9			
14	To begin a whole new problem go to step 2			

Program Listings

Input

```
01♦LBL "CON
DUIT"
02 SF 21
03 SF 27
04 "U=?"
05 PROMPT
06 STO 09
07 "RHO=?"
08 PROMPT
09 STO 10
10 ST/ 09
11 "E=?"
12 PROMPT
13 STO 14
14 "L=?"
15 PROMPT
16 STO 03
17 "D=?"
18 PROMPT
19 STO 13
20 "KT=?"
21 PROMPT
22 4
23 /
24 STO 08
25♦LBL C
26 CF 22
27 "V=?"
28 PROMPT
29 SF 00
30 FS? 22
31 CF 00
32 STO 02
33 "DP=?"
34 PROMPT
```

1st V

```
35 STO 04
36 XEQ 09
37 FS? 00
38 GTO 03
```

Calculate ΔP

```
39 RCL 02
40 X↑2
41 *
42 RCL 10
43 *
44 STO 04
45 "DP="
46 GTO 10
```

Iterate to find V using 1st V as guess

```
47♦LBL 03
48 RND
49 STO 00
50 XEQ 08
51 RND
52 RCL 00
53 X<>Y
54 X≠Y?
55 GTO 03
56 "V=?"
57 RCL 02
58 GTO 10
```

Calculate constants

```
59♦LBL 09
60 RCL 10
61 RCL 13
62 RCL 14
63 /
64 STO 06
65 LN
66 1.737
67 STO 07
68 *
69 2.28
70 +
71 STO 12
72 STO 05
73 FS? 00
74 GTO 07
```

Is flow turbulent?

```
75♦LBL 08
76 16
77 RCL 02
78 RCL 13
79 *
80 RCL 09
81 /
82 STO 01
83 2300
84 X<=Y?
85 GTO 02
86 RDN
87 /
88 SQRT
89 1/X
90 STO 05
91 GTO 07
```

Iterate to find $\frac{1}{\sqrt{f}}$

```
92♦LBL 02
93 RCL 12
94 RCL 05
95 -
96 4.67
97 RCL 06
98 *
99 RCL 01
```

Program Listings

```
/
RCL 05
*
1
+
STO 11
LN
RCL 07
*
-
RCL 11
1/X
CHS
1
+
RCL 07
*
RCL 05
/
1
+
/
ST+ 05
RCL 05
/
ABS
1 E-3
X<=Y?
GTO 02
◆LBL 07
RCL 05
1/X
X↑2
RCL 03
*
RCL 13
/
RCL 08
+
2
*
RCL 04
RCL 10
/
X<>Y
FS? 00
GTO 00
RTN
◆LBL 00
/
```

```
SQRT
STO 02
RTN
◆LBL 10
ARCL X
AVIEW
"Re="
ARCL 01
AVIEW
"F="
RCL 05
1/X
X↑2
ARCL X
AVIEW
END
```

Output

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	V	50	
	Re		
	V		
	L		
	ΔP		
05	$1/\sqrt{f}$	55	
	ρ/ε		
	1.737		
	$K_T/4$		
	μ		
10	ρ	60	
	used		
	$1/\sqrt{f_0}$		
	D		
	ε		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 015 TOT. REG. 52 USER MODE
ENG ____ FIX ____ SCI ____ ON x OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		calculate V	calculate ΔP
22		calculate ΔP	calculate V
21	S	printer enable	printer disable
27	S	user mode on	user mode off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
change v or ΔP	C		

CONDUIT FLOW

PROGRAM REGISTERS NEEDED: 38

ROW 1 (1 : 2)

ROW 2 (3 : 7)

ROW 3 (7 : 14)

ROW 4 (14 : 20)

ROW 5 (20 : 27)

ROW 6 (27 : 33)

ROW 7 (33 : 41)

ROW 8 (42 : 50)

ROW 9 (50 : 57)

ROW 10 (58 : 66)

ROW 11 (66 : 74)

ROW 12 (74 : 83)

ROW 13 (83 : 93)

ROW 14 (94 : 103)

ROW 15 (104 : 116)

ROW 16 (117 : 126)

ROW 17 (126 : 137)

ROW 18 (138 : 148)

CONDUIT FLOW

ROW 19 (149 : 157)

ROW 20 (157 : 165)

ROW 21 (165 : 165)

FLOW WITH A FREE SURFACE

This program solves algebraic manipulations of the following two equations for any of the five variables in each:

Manning flow formula:

$$S = \frac{(nQ)^2}{2.2082\ r^{4/3} \times a^2} \qquad (1)$$

$$Q = \frac{K}{n} D^{8/3} \times S^{1/2} \qquad (2)$$

Where

S = slope of the bottom, dimensionless

n = roughness coefficient

r = hydraulic radius ft.

Q = discharge rate ft^3/sec

a = crosssection area ft^2/sec

K = discharge factor dimensionless

and D = hydraulic diameter

References: Civil Engineering Handbook, Leonard Church Urquhart (ed.), McGraw-Hill Book Company, 4th Ed.
HP-67/HP-97 Users' Library program #00269D.

Example: 1. Find Q for S = .001, N = .013, R = 5/12, and A = 5.0

2. Find K for S = .001, n = .014, Q = 200, and D = 4

Solution:

Keystrokes:	Display:	
[USER]		**(Set USER mode)**
[XEQ] [ALPHA] SIZE [ALPHA] 007		
[XEQ] [ALPHA] FSFLO [ALPHA]	LBL A OR B ?	
[A]	S ?	
.001 [R/S]	N ?	
.013 [R/S]	Q ?	
[R/S]	R ?	
5 [ENT] 12 [÷] [R/S]	A ?	
5 [R/S]	Q=10.0826	
[B]	S ?	
.001 [R/S]	N ?	
.014 [R/S]	Q ?	
200 [R/S]	D ?	
4 [R/S]	K ?	
[R/S]	K=2.1962	

User Instructions

				SIZE: 007
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program and set USER mode		[USER]	
2	Initialize the program		[XEQ] FSFLO	LBL A OR B ?
3	For Manning Flow formula, press →		[A]	S ?
	and input 4 of the following:	S	[R/S]	N ?
	When prompted for the unknown variable,	N	[R/S]	Q ?
	press [R/S] (make no input). The unknown	Q	[R/S]	R ?
	is then automatically calculated.	R	[R/S]	A ?
		A	[R/S]	S=( )
				-or- N=( )
				-or- Q=( )
				-or- R=( )
				-or- A=( )
4	For discharge factor formula, press →		[B]	S ?
	and input 4 of the following:		[R/S]	N ?
	When prompted for the unknown variable,		[R/S]	Q ?
	press [R/S] (make no input). The		[R/S]	D ?
	unknown is then automatically calculated.		[R/S]	K ?
			[R/S]	S=( )
				-or- N=( )
				-or- Q=( )
				-or- D=( )
				-or- K=( )

Program Listings

```
01♦LBL "FSFLO"
02 "LBL A OR B?"
03 PROMPT
```

Equation 1

```
04♦LBL A
05 1.1
06 STO 00
07 CF 22
```

Prompt and store data

```
08 "S ?"
09 XEQ 11
10 "N ?"
11 XEQ 11
12 "Q ?"
13 XEQ 11
14 "R ?"
15 XEQ 11
16 "A ?"
17 XEQ 11
18 RCL 04
19 4
20 ENTER↑
21 3
22 /
23 Y↑X
24 2.2082
25 *
26 STO 04
27 GTO IND 06
```

Calculate S

```
28♦LBL 01
29 XEQ 16
30 RCL 01
31 *
32 XEQ 15
33 /
34 "S"
```

Display routine

```
35♦LBL 12
36 "⊢="
37 ARCL X
38 PROMPT
```

Input storage routine

```
39♦LBL 11
40 PROMPT
41 STO IND 00
42 RCL 00
43 FC?C 22
44 STO 06
45 ISG 00
46 RTN
```

Calculate $\frac{(NQ)^2}{S}$

```
47♦LBL 16
48 RCL 02
49 RCL 03
50 *
51 X↑2
52 RCL 01
53 /
54 RTN
```

Calculate denominator

```
55♦LBL 15
56 RCL 05
57 X↑2
58 RCL 04
59 *
60 RTN
```

Calculate N

```
61♦LBL 02
62 XEQ 15
63 RCL 01
64 *
65 SQRT
66 RCL 03
67 /
68 "N"
69 XEQ 12
```

Calculate Q

```
70♦LBL 03
71 XEQ 15
72 RCL 01
73 *
74 SQRT
75 RCL 02
76 /
77 "Q"
78 XEQ 12
```

Calculate A

```
79♦LBL 05
80 XEQ 16
81 RCL 04
82 /
83 SQRT
84 "A"
85 XEQ 12
```

Calculate R

```
86♦LBL 04
87 XEQ 16
88 RCL 05
89 X↑2
90 /
91 2.2082
92 /
93 3
94 ENTER↑
95 4
96 /
97 Y↑X
98 "R"
```

Program Listings

```
 99 XEQ 12
100◆LBL B          ----------------
                   Equation 2
101 1.1
102 STO 00
103 CF 22
104 "S ?"
105 XEQ 11         Prompt and
106 "N ?"          store date
107 XEQ 11
108 "Q ?"
109 XEQ 11
110 "D ?"
111 XEQ 11
112 "K ?"
113 XEQ 11
114 5              adjust pointer
115 ST+ 06
116 RCL 01
117 SQRT
118 STO 01
119 RCL 04
120 8
121 ENTER↑
122 3
123 /
124 Y↑X
125 STO 04         D^(8/3)
126 GTO IND
06
                   ----------------
127◆LBL 08
128 RCL 05
129 RCL 02         Calculate Q
130 /
131 RCL 04
132 *
133 RCL 01
134 *
135 "Q"
136 XEQ 12
                   ----------------
137◆LBL 06
138 RCL 03
139 RCL 05         Calculate S
140 RCL 02
141 /
142 RCL 04
143 *
144 /
145 X↑2
146 "S"
147 XEQ 12
                   ----------------
148◆LBL 10
```

```
149 RCL 02         Calculate K
150 RCL 03
151 RCL 01
152 /
153 *
154 RCL 04
155 /
156 "K"
157 XEQ 12
                   ----------------
158◆LBL 07
159 RCL 05         Calculate N
160 RCL 04
161 *
162 RCL 03
163 RCL 01
164 /
165 /
166 "N"
167 XEQ 12
                   ----------------
168◆LBL 09
169 RCL 03         Calculate D
170 RCL 01
171 /
172 RCL 02
173 *
174 RCL 05
175 /
176 3
177 ENTER↑
178 8
179 /
180 Y↑X
181 "D"
182 XEQ 12
183 .END.
```

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Pointer	50	
	S or NS		
	n		
	Q		
	2.2082 $r^{4/3}$ or $D^{8/3}$		
05	a or k	55	
	subroutine pointer		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 053 USER MODE

ENG ___ FIX 4 SCI ___ ON x OFF ___

DEG x RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FLOW WITH A FREE SURFACE

PROGRAM REGISTERS NEEDED: 47

ROW 1 (1 - 2)

ROW 2 (2 - 5)

ROW 3 (5 - 10)

ROW 4 (10 - 14)

RCW 5 (14 - 17)

ROW 6 (18 - 25)

ROW 7 (26 - 33)

ROW 8 (34 - 41)

ROW 9 (42 - 51)

ROW 10 (52 - 62)

ROW 11 (62 - 71)

ROW 12 (71 - 79)

ROW 13 (80 - 87)

ROW 14 (87 - 93)

ROW 15 (94 - 101)

ROW 16 (101 - 106)

ROW 17 (106 - 110)

ROW 18 (110 - 113)

FLOW WITH A FREE SURFACE

ROW 19 (114 – 125)

ROW 20 (126 – 136)

ROW 21 (136 – 146)

ROW 22 (147 – 156)

ROW 23 (157 – 166)

ROW 24 (167 – 177)

ROW 25 (178 – 183)

PIPE SLIDE-RULE

The program computes the unknown, when given the surface coefficient (n) and any three of the following: 1) Flow (Q); 2) Slope (S); 3) Pipe diameter (D); 4) Depth % (D/d). Also computes Velocity (V). When solving for Pipe diameter, the program automatically rounds up to a standard size of 6", 8", 12", 15", 18", 21", 24", etc.--pipe. Depth percentage (D/d) is found by Newton's method of iteration. In the case of depth percentages between approximately 82% and 100%, two roots or values are appropriate.

Reference: HP-67/97 Users' Library program #00281D by C. B. Coleman.

Example: Find D/d and V given the following:

n = .013 s = .001 (1ft/1000ft)*

Q = 850.3 CFS D = 144 (inches)

Solution:

Keystrokes:		Display:
[///] [FIX] 2		
[XEQ] [ALPHA] SIZE [ALPHA] 013		
[XEQ] [ALPHA] PSR [ALPHA] [A]		N ?
.013 [R/S]		S ?
.001 [R/S]		Q ?
850.3 [R/S]		D ?
144 [R/S]		D/d ?
[R/S]		D/d=81.96
[R/S]	(second solution)	D/d=100.00
[B]		Q ?
850.3 [R/S]		D ?
144 [R/S]		D/d ?
81.96 [R/S]		V=8.57
etc.	(for second solution	V=7.52)

*Note that slope is given as rise/run, not %. A 30% slope would be entered as .30.

User Instructions

SIZE: 013

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program.			
2	Initialize		[XEQ] PSR	
3	To solve for S, Q, D or D/d:		[A]	N ?
4	Input the surface coefficient:	n	[R/S]	S ?
5	Input any three of the following:			
	Slope (ft rise/ft run; not as a percent)	S	[R/S]	Q ?
	Flow (ft^3/sec)	Q	[R/S]	D ?
	Pipe Diameter (inches)	D	[R/S]	D/d ?
	Depth Percent (%)	D/d	[R/S]	Q=( )
				or S=( )
				or D=( )
				or D/d=( )
	When prompted for the unknown variable,			
	press [R/S] (make no input). When solving			
	for D/d, press [R/S] a second time (if no			
	printer is present) to find the second			
	solution.			
6	To calculate velocity:		[B]	Q ?
	Input: Flow (ft^3/sec)	Q	[R/S]	D ?
	Pipe Diameter (inches)	D	[R/S]	D/d ?
	Depth Percent (%)	D/d	[R/S]	V=( )
	Velocity is in ft/sec.			

Program Listings

```
01♦LBL "PSR
"
02 SF 21          Initialize
03 SF 27
04 RAD
05 STOP
06♦LBL B          Solve for V
07 SF 06
08 10
09 STO 04
10 XEQ 05         Input Routine
11 XEQ 02
12 ENTER↑
13 SIN
14 -
15 1/X
16 RCL 10         Calculate V
17 *
18 RCL 11
19 X↑2
20 /
21 1152
22 *
23 "V"
24 GTO 14         Output Routine
25♦LBL A
26 CF 06
27 8
28 STO 04
29 "N"
30 XEQ 00
31 "S"            Input Knowns
32 XEQ 00
33♦LBL 05
34 "Q"
35 XEQ 00
36 "D"
37 XEQ 00
38 "D/d"
39 XEQ 00
40 FS?C 06
41 RTN
42 100
43 ST* 09
44 GTO IND
03
45♦LBL 00
46 "⊢ ?"
47 PROMPT
48 STO IND        Common Input
04                Routine
49 RCL 04
50 FC?C 22
51 STO 03
52 ISG 04
53 CLD
54 RTN
55♦LBL 10         Solve for Q
56 SF 05
57 XEQ 02
58 XEQ 15
59 "Q"            Output Q
60 GTO 14
61♦LBL 09
62 1
63 STO 09         Solve for S
64 XEQ 20
65 X↑2
66 100
67 /
68 "S"            Output S
69 GTO 14
70♦LBL 11
71 1
72 STO 11
73 XEQ 20
74 .375
75 Y↑X
76 6
77 X<>Y
78 X<=Y?
79 GTO 04
80 8              Solve for D
81 X<>Y
82 X<=Y?
83 GTO 04         Rounding to
84 12             Standard Size
85 X<>Y
86 X<=Y?
87 GTO 04
88 3
89 /
90 .999
91 +
92 INT
93 3
```

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

Program Listings

```
 94 *
 95 X<>Y
 96♦LBL 04
 97 X<>Y
 98 "D"              Output D
 99 GTO 14
100♦LBL 02
101 RCL 12
102 50               Common Routine
103 /
104 1
105 -
106 CHS
107 ACOS
108 ST+ X
109 STO 12
110 RTN
111♦LBL 12
112 PI               Solve for D/d
113 STO 12
114 4
115 STO 01
116 CHS
117 STO 02
118♦LBL 15
119 RCL 11           Common
120 8                Iteration
121 ENTER↑           Routine
122 3
123 /
124 Y↑X
125 RCL 09
126 SQRT
127 *
128 RCL 08
129 /
130 320
131 X↑2
132 /
133 STO 05
134♦LBL 01
135 RCL 12
136 ENTER↑
137 SIN
138 -
139 5
140 Y↑X
141 RCL 12
142 X↑2
143 /
144 3
145 1/X
146 Y↑X
147 RCL 05
148 *
149 FS?C 05
150 RTN
151 RCL 10
152 -
153 STO 00
154 RCL 01
155 RCL 12
156 STO 01
157 -
158 RCL 02
159 RCL 00
160 STO 02
161 -
162 /
163 *
164 CHS
165 ST+ 12
166 RCL 12
167 /
168 RND
169 X≠0?
170 GTO 01
171 7
172 RCL 12
173 X>Y?
174 GTO 17
175 2
176 /
177 COS
178 1
179 -
180 2
181 /
182 CHS
183 1 E2
184 *
185 STO 06
186 .8196
187 X<>Y
188 "D/d"
189 X<=Y?
```

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

Program Listings

```
190 GTO 14          Output D/d
191 FC? 07
192 XEQ 14
193 FS?C 07
194 GTO 14
195 SF 07
196 RCL 12
197 STO 07
198 PI
199 ST+ X
200 STO 01
201 CHS
202 STO 02
203 6
204 STO 12
205 GTO 15
206♦LBL 20
207 SF 05
208 XEQ 02
209 XEQ 15
210 RCL 10
211 /
212 1/X
213 RTN
214♦LBL 14
215 "⊢="             Output Routine
216 ARCL X
217 AVIEW
218 END
```

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	used	50	
	used		
	used		
	subroutine pointer		
	pointer		
05	used	55	
	used		
	used		
	n		
	S		
10	Q	60	
	D		
	D/d		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 013 TOT. REG. 60 USER MODE

ENG ___ FIX ___ SCI ___ ON X OFF ___

DEG ___ RAD X GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
05	C	Calculating Q, S, or D	Calculating D/d
06	C	Calculating V	Calculating any other
07	C	Second D/d solution	First or only D/d
21	S	Printer Enabled	Printer Disabled
22	C	Data input via keyboard	No data input
27	S	USER on	USER off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Input for Q, S D or D/d	A		
Input for V	B		

PIPE SLIDE RULE

PROGRAM REGISTERS NEEDED: 48

HEWLETT PACKARD
SOLUTIONS BOOK
FLUID DYNAMICS/HYDRAULICS

ROW 1 (1 : 5)

ROW 2 (6 : 11)

ROW 3 (12 : 21)

ROW 4 (22 : 29)

ROW 5 (30 : 35)

ROW 6 (35 : 39)

ROW 7 (40 : 46)

ROW 8 (46 : 54)

ROW 9 (55 : 60)

ROW 10 (61 : 68)

ROW 11 (69 : 75)

ROW 12 (76 : 85)

ROW 13 (86 : 94)

ROW 14 (95 : 104)

ROW 15 (105 : 116)

ROW 16 (117 : 128)

ROW 17 (129 : 139)

ROW 18 (140 : 151)

ROW 19 (152 : 164)

ROW 20 (165 : 174)

ROW 21 (174 : 184)

ROW 22 (185 : 190)

ROW 23 (190 : 196)

ROW 24 (197 : 206)

ROW 25 (206 : 213)

ROW 26 (214 : 218)

FORCES AT BENDS AND FITTINGS

When the velocity of a flowing fluid is changed in magnitude or direction, such as at a bend or fitting, a force must act upon the fluid to cause the change. This program considers only the systems where the pipe itself is pressure-tight, but where accelerating forces must be resisted by external means to prevent movement of the piping, increased stress in the pipe walls, or both. The equations are:

$$F_x = \frac{QW}{g}(V_{2x} - V_{1x}), \qquad F_y = \frac{QW}{g}(V_{2y} - V_{1y})$$

$$\overline{F} = \vec{F}_x + \vec{F}_y , \qquad \overline{R} = -\overline{F} ,$$

where $\overline{F}$ = accelerating force on water, lbs

Q = rate of flow, ft^3/sec

W = specific weight, lbs/ft^3

g = acceleration of gravity, ft/sec^2

V_2 = velocity leaving fittings, ft/sec

V_1 = velocity entering fitting, ft/sec

$\overline{R}$ = reaction of water on fitting, lbs

Subscripts: x for x direction and y for y direction

References: Fluid Mechanics and Hydraulics, by Ronald V. Giles, Schaums Outline Series, McGraw-Hill B-ok Company, New York, 1962.
Hp-67/HP-97 Users' Library program #00268D

Example:



$Q = 2\ ft.^3/sec.$

$A_1 = 0.2\ ft.^2$

$A_2 = 0.016\ ft.^2$

Find the forces on the water deck gun due to the changes in velocity and direction.

$V,x_1 = 0$

$V,x_2 = V\cos\theta;\ \theta = 30°,\ V = \frac{2}{.016}\ \left(\frac{Q}{A_2}\right).$

$V,y_1 = 2/.2 \quad (Q/A_1)$

$V,y_2 = V\sin\theta;\ \theta = 30°,\ V = \frac{2}{.016}\ \left(\frac{Q}{A_2}\right).$

Solution:

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 002	
[XEQ] [ALPHA] FORCE [ALPHA]	W ?
62.4 [R/S]	Q ?
2 [R/S]	V,X1 ?
0 [R/S]	V,X2 ?
108.25 [R/S]	V,Y1 ?
10 [R/S]	V,Y2 ?
62.5 [R/S]	F,X=419.9∡0.0
[R/S]	F,Y=203.6∡90.0
[R/S]	F=466.7∡25.9
[R/S]	R=466.7∡-154.1

User Instructions

SIZE: 002

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program			
2	Initialize the program		[XEQ] FORCE	W ?
3	Input: specific weight of fluid	W	[R/S]	Q ?
	flow rate	Q	[R/S]	V,X1 ?
	x-component of velocities: entering	V,X1	[R/S]	V,X2 ?
	leaving	V,X2	[R/S]	V,Y1 ?
	y-component of velocities: entering	V,Y1	[R/S]	V,Y1 ?
	leaving	V,Y2	[R/S]	
4	Find: x-component of force;			F,X=( )∡0.00
	y-component of force;			F,Y=( )∡90.0
	net force; and			F=( )∡( )
	net reaction.			R=( )∡( )

Program Listings

```
01♦LBL "FOR
CE"
02 "W ?"
03 PROMPT
04 "Q ?"
05 PROMPT
06 *
07 32.1739              WQ/g
08 /
09 STO 00
10 STO 01
11 "V,X1 ?"
12 PROMPT
13 "V,X2 ?"
14 PROMPT
15 -
16 CHS
17 ST* 00               Fx
18 "V,Y1 ?"
19 PROMPT
20 "V,Y2 ?"
21 PROMPT
22 -
23 CHS
24 ST* 01               Fy
25 0
26 RCL 00
27 "F,X"
28 XEQ 08
29 90
30 RCL 01
31 "F,Y"
32 XEQ 08
33 RCL 00
34 R-P
35 "F"
36 XEQ 08
37 RCL 01
38 CHS
39 RCL 00
40 CHS
41 R-P
42 "R"
                     ----------------
43♦LBL 08            Display routine
44 "├="
45 ARCL X
46 "├∡"
47 ARCL Y
48 PROMPT
49 RTN
50 .END.
```

51	
60	
70	
80	
90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	QW/g, FX	50	
	QW/g, FY		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 002 TOT. REG. 019 USER MODE

ENG ___ FIX 1 SCI ___ ON ___ OFF x

DEG x RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FORCES AT BENDS AND FITTINGS

PROGRAM REGISTERS NEEDED: 17

ROW 1 (1 - 2)

ROW 2 (3 - 7)

ROW 3 (7 - 13)

ROW 4 (13 - 18)

ROW 5 (18 - 20)

ROW 6 (21 - 28)

ROW 7 (28 - 34)

ROW 8 (35 - 43)

ROW 9 (44 - 50)

ROW 10 (50 - 50)

VALVE SIZING

This program calculates the valve coefficient Cv.

Valve for Liquid flow:

$$Cv = Q\sqrt{\frac{G}{(P1-P2)}}$$

Valve for Gas flow, if $P2 \geq \frac{P1}{2}$:

$$Cv = \frac{Q\sqrt{G\ Ta}}{1360\ \sqrt{(P1-P2)P2}}$$

Valve for Gas flow, if $P2 < \frac{P1}{2}$:

$$Cv = \frac{Q\sqrt{G\ Ta}}{1360\ (P1/2)}$$

Valve for Vapor flow, if $P2 \geq \frac{P1}{2}$:

$$Cv = \frac{Q}{63.4}\sqrt{\frac{Vs}{(P1-P2)}}$$

Valve for Vapor flow, if $P2 < \frac{P1}{2}$:

$$Cv = \frac{Q}{63.4}\sqrt{\frac{Vs}{(P1/2)}}$$

Valve for Steam flow, if $P2 \geq \frac{P1}{2}$:

$$Cv = \frac{Q\ (1+0.0007\ Ts)}{3\ \sqrt{(P1-P2)P2}}$$

Valve for Steam flow, if $P2 < \frac{P1}{2}$:

$$Cv = \frac{Q\ (1+0.0007\ Ts)}{3\ (P1/2)}$$

References: HP-67/HP-97 Users' Library program #02200D by Paul Crowder.
"Process Instruments and Control Handbook" by D. M. Constadine, pub. McGraw-Hill, 1957.

Example: Calculate CV for a gas valve:

Inlet pressure - 135 psig
Outlet pressure - 115 psig
Sp. Gr. @ flow - 1.0

Flow temp. - 10°F
Flow rate - 9000 CFH

Solution:

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 004	
[XEQ] CV	L,G,V, OR S ?
[ALPHA] G [ALPHA] [R/S]	Q ?
9000 [R/S]	P1 ?
149.7 [R/S]	P2 ?
129.7 [R/S]	G ?
1 [R/S]	T ?
10 [R/S]	CV = 2.82

User Instructions

SIZE: 004

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize the program		[XEQ] CV	L,G,V,or S?
3	Select one: liquid	"L"	[R/S]	Q ?
	gas;	"G"	[R/S]	Q ?
	vapor; or	"V"	[R/S]	Q ?
	steam	"S"	[R/S]	Q ?
4	Input only values prompted for:			
	Flow (liq.-GPM; gas-CFH; vap/stm-#/hr)	Q	[R/S]	P1 ?
	Inlet pressure (PSIA=PSIG+14)	P1	[R/S]	P2 ?
	Outlet pressure (PSIA)	P2	[R/S]	(prompt)
	Specific gravity at flow	G	[R/S]	(prompt)
	Temperature (°F)	Ts	[R/S]	(prompt)
	Specific volume (ft^3/lb)	Vs	[R/S]	(prompt)
5	CV, the valve coefficient, results			CV=( )

Program Listings

Program	Comments
01♦LBL "CV"	
02 CF 01	
03 "L,G,V, OR S?"	
04 PROMPT	Select equation
05 ASTO Y	
06 "L"	
07 ASTO X	
08 X=Y?	
09 GTO 01	
10 "G"	
11 ASTO X	
12 X=Y?	
13 GTO 02	
14 "V"	
15 ASTO X	
16 X=Y?	
17 GTO 03	
18 XEQ 05	Calculate CV for steam
19 "TS ?"	
20 PROMPT	
21 .0007	
22 *	
23 1	
24 +	
25 3	
26 /	
27 RCL 00	
28 *	
29 XEQ 06	
30 FS?C 01	
31 GTO 08	
32♦LBL 10	$P_2 \geq \frac{P_1}{2}$
33 RCL 01	
34 RCL 02	
35 *	
36 SQRT	
37 /	
38 XEQ 09	
39♦LBL 08	$P_2 < \frac{P_1}{2}$
40 RCL 01	
41 /	
42♦LBL 09	Display routine
43 "CV="	
44 ARCL X	
45 PROMPT	
46♦LBL 03	Calculate CV for vapor
47 XEQ 05	
48 RCL 00	
49 63.4	
50 /	
51 XEQ 06	
52 "VS ?"	
53 PROMPT	
54 RCL 01	
55 /	
56 SQRT	
57 *	
58 CF 01	
59 XEQ 09	
60♦LBL 02	Calculate CV for gas
61 XEQ 05	
62 RCL 00	
63 1360	
64 /	
65 "G ?"	
66 PROMPT	
67 "T ?"	
68 PROMPT	
69 460	absolute temp.
70 +	
71 *	
72 SQRT	
73 *	
74 XEQ 06	
75 FS?C 01	
76 GTO 08	
77 GTO 10	
78♦LBL 01	Calculate CV for liquid
79 XEQ 05	
80 "G ?"	
81 PROMPT	
82 RCL 01	
83 RCL 02	
84 -	
85 /	
86 SQRT	
87 RCL 00	
88 *	
89 XEQ 09	
90♦LBL 05	Prompt and store input
91 "Q ?"	
92 PROMPT	
93 STO 00	
94 "P1 ?"	
95 PROMPT	
96 STO 01	
97 "P2 ?"	
98 PROMPT	
99 STO 02	
100 RTN	
101♦LBL 06	

Program Listings

Program	Comments
102 STO 03	test P_1, P_2
103 RCL 02	$\frac{P_1}{2}$
104 RCL 01	
105 2	
106 /	
107 X>Y?	
108 GTO 07	
109 RCL 02	
110 ST- 01	$P_2 \geq \frac{P_1}{2}$
111 RCL 03	
112 RTN	
113♦LBL 07	
114 STO 01	
115 SF 01	$P_2 < \frac{P_1}{2}$
116 RCL 03	
117 RTN	
118 .END.	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Q	50	
	P1, P1/2 or P1-P2		
	P2		
	Used		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 004 TOT. REG. 035 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF x

DEG x RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01	C	P2 < P1/2	P2 ≥ P1/2

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

VALVE SIZING

PROGRAM REGISTERS NEEDED: 32

ROW 1 (1 – 3)

ROW 2 (3 – 6)

ROW 3 (7 – 14)

ROW 4 (14 – 19)

ROW 5 (19 – 27)

ROW 6 (28 – 36)

ROW 7 (37 – 44)

ROW 8 (44 – 51)

ROW 9 (51 – 58)

ROW 10 (58 – 63)

ROW 11 (64 – 69)

ROW 12 (69 – 77)

ROW 13 (77 – 84)

ROW 14 (85 – 92)

ROW 15 (93 – 97)

ROW 16 (98 – 109)

ROW 17 (110 – 118)

PIPE NETWORK ANALYSIS

(Requires at least one additional memory module)

The first portion of the program computes the equivalent length of a pipe by use of the Hazen-Williams equation. This is a prerequisite to any analysis of a water distribution system by the version of the Hardy-Cross Method used herein which requires that all pipes must be of the same diameter d (normally d=10" is used) and roughness C (normally C=100 is used.)

This portion of the program computes the equivalent length by the equation:

$$L_2 = L_1 \left(\frac{C_2}{C_1}\right)^{1.8519} \cdot \left(\frac{d_2}{d_1}\right)^{4.8707}$$

where the subscript 1 represents the existing pipe and the subscript 2 the equivalent pipe.

The second portion of the program computes corrected flows (4 iterations) using the Hardy-Cross method derived here:

Consider a two pipe loop; flow in one pipe Q_1 and the second pipe Q_2.

1) Head Loss (in one pipe) = h = KLQ^n where K is a constant dependent on the diameter, roughness and length of the pipe.

$$K = \left(\frac{1.594}{C}\right)^{1.85} \cdot \frac{L}{D}4.87$$

2) Balanced head losses (H_1 and $-H_2$) around loop are equal to 0

$$\Sigma H = H_1 - H_2 = 0$$

3) If the assumed flow split Q_1 and $-Q_2$ are each in error by the same amount ΔQ

$$\Sigma H = \Sigma KL \ (Q + \Delta Q)^n = 0$$

4) Expanding the polynomial and neglecting all but the first two terms

$$\Sigma H = \Sigma KLQ^n + \Sigma nKL\Delta Q \cdot Q^{n-1} = 0 \text{ whence } \Delta Q = \frac{-\Sigma KLQ^n}{n\Sigma KLQ^{n-1}} = \frac{-\Sigma KLQ^n}{n\Sigma KL\frac{Q^n}{Q}}$$

5) However, since all pipes are of the same size and roughness, K cancels. Also n = 1.85 (=1/0.54)

$$\Delta Q = \frac{-\Sigma LQ^{1/0.54}}{\sum \left(\frac{LQ^{1/0.54}}{Q(0.54)}\right)}$$

The third portion of the program enables the computation of the pressure (in psi) at any pipe junction in the distribution system given the starting pressure at a particular pipe junction (normally a point where flow enters the system from a pump or elevated tank). This portion of the program computes the head loss in feet along a particular pipe by the formula

$$h_f = L \left[\left(\frac{3.5521Q}{C} \right)^{1.8519} \cdot d^{-4.8704} \right]$$

Notes:

1. All pipes must be assumed to carry some flow, no matter how small. No zero flows.

2. The program assumes loops comprise 4 pipes. If a loop contains 3 pipes, enter a zero for 4th pipe number.

3. Prior to running the program, sketch the pipe network and assign pipe numbers and flow directions (clockwise or counterclockwise). Adjacent loops must have opposite flow directions to keep the flow consistant in common pipes.



For example, the flow direction in pipe 3 (above) is the same for both loops. Then assign assumed flows in the direction of the arrows positive values and assumed flows opposite the arrows negative values. If the assumed flow in a particular pipe is inadvertently assumed in the wrong direction, the program will automatically correct the sign during the program iterations.

4. SIZE = No. of loops + 2 x No. of pipes + 7.

5. The annunciators in the display show the progress of the calculations (flags 1-4 correspond to iterations 1-4).

Reference: HP-67/HP-97 USERS LIBRARY program #02275D by Bernard Golding.

Example:

Compute the correct flows for the network below. Also, find the pressure at point D if the pressure at point A is 100 psi.



Solution:

Keystrokes:	Display:
[USER]	(Set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 023	
[XEQ] [ALPHA] NA [ALPHA]	PIPE NO. ?
1 [R/S]	PIPE NO. ?
2 [R/S]	PIPE NO. ?
3 [R/S]	PIPE NO. ?
4 [R/S]	A OR B ?
[A]	PIPE NO. ?
3 [R/S]	PIPE NO. ?
5 [R/S]	PIPE NO. ?
6 [R/S]	PIPE NO. ?
7 [R/S]	A OR B ?

[B]	NO. OF PIPES ?
7 [R/S]	EQUIV. DIA. ?
10 [R/S]	EQUIV. RUF. ?
100 [R/S]	L,1 ?
2000 [R/S]	d,1 ?
12 [R/S]	C,1 ?
100 [R/S]	Q,1 ?
3000 [R/S]	L,2 ?
2000 [R/S]	d,2 ?
10 [R/S]	C,2 ?
100 [R/S]	Q,2 ?
2000 [R/S]	L,3 ?
2000 [R/S]	d,3 ?
14 [R/S]	C,3 ?
100 [R/S]	Q,3 ?
6250 [CHS] [R/S]	L,4 ?
2000 [R/S]	d,4 ?
16 [R/S]	C,4 ?
100[R/S]	Q,4 ?
9750 [CHS] [R/S]	L,5 ?
2000 [R/S]	d,5 ?
12 [R/S]	C,5 ?
100 [R/S]	Q,5 ?
1500 [CHS] [R/S]	L,6 ?
2000 [R/S]	d,6 ?
12 [R/S]	C,6 ?
100 [R/S]	Q,6 ?
250 [R/S]	L,7 ?
2000 [R/S]	d,7 ?
14 [R/S]	C,7 ?
100 [R/S]	Q,7 ?
1000 [R/S]	Q,1=3404
[R/S]	Q,2=2404
[R/S]	Q,3=-4134
[R/S]	Q,4=-9346

[R/S]	Q,5=212
[R/S]	Q,6=1962
[R/S]	Q,7=2712
[C]	EQUIV. DIA. ?
10 [R/S]	EQUIV. RUF. ?
100 [R/S]	dP,1=34.55
[R/S]	dP,2=44.09
[R/S]	dP,3=23.37
[R/S]	dP,4=55.24
[R/S]	dP,5=0.20
[R/S]	dP,6=12.45
[R/S]	dP,7=10.71

Since the pressure drops in the direction of the flow (and increases moving upstream), the pressure at point D=100.00 - 55.24 - 23.37 + .20 = 21.59psi.

User Instructions

				SIZE: 007+
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program and set USER mode		[USER]	
2	Initialize the program		[XEQ] NA	PIPE NO. ?
3	Input 4 pipe numbers for each loop.	N	[R/S]	PIPE NO. ?
	Enter a zero for pipe number when there	N	[R/S]	PIPE NO. ?
	are less than 4 pipes in a loop.	N	[R/S]	PIPE NO. ?
		N	[R/S]	A OR B ?
4	For another pipe loop, press		[A]	PIPE NO. ?
	and go to step 3			
5	When all pipe loops are in, press		[B]	NO. OF PIPES ?
6	Input: the number of pipes;	n	[R/S]	EQUIV. DIA. ?
	equivalent diameter;	d	[R/S]	EQUIV. RUF. ?
	and equivalent roughness	C	[R/S]	L,1 ?
7	Input: pipe length;	L,x	[R/S]	d,x ?
	pipe diameter;	d,x	[R/S]	C,x ?
	pipe roughness;	C,x	[R/S]	Q,x ?
	and assumed flow.	±Q,x	[R/S]	L,x+1 ?
				or Q,1=C
8	If a mistake is made in putting L,d, or C,			
	press		[GTO] 03	
	and go to step 7		[R/S]	L,x ?
9	Perform step 7 until the data for all pipes			
	is input, at which time the program com-			
	putes corrected flows. If a printer is in			
	the system, the ΔQ's will be printed that			
	the user may watch convergence.			
10	To see the corrected flows, press		[R/S]	Q,x=( )
11	Repeat step 9 until x=n			

User Instructions

SIZE: 007+

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
12	To find the pressure drops along a			
	pipe, press		[C]	EQUIV. DIA. ?
13	Input: equivalent diameter;	d	[R/S]	EQUIV. RUF. ?
	and equivalent roughness.	C	[R/S]	dP,1=( )
14	To see the ΔP's, press		[R/S]	dP,x=( )
	and repeat until x=n			

Program Listings

```
 01♦LBL "NA"
 02 FIX 0
 03 CF 29
 04 CF 01              Initialize
 05 CF 02
 06 CF 03
 07 CF 04
 08 6.3
 09 STO 03
                       ------------------
 10♦LBL A
 11 ISG 03             Set up control
 12 RCL 03             registers
 13 STO 00
 14 .01
 15 STO 04
 16 0
 17 STO IND
03
 18♦LBL 01
 19 "PIPE NO
  ?"
 20 PROMPT
 21 RCL 04
 22 *
 23 ST+ IND
03
 24 .01
 25 ST* 04
 26 RCL 04
 27 1 E-9
 28 X<=Y?              4th pipe?
 29 GTO 01             no
 30 "A OR B
?"
 31 PROMPT
                       ------------------
 32♦LBL B
 33 RCL 03
 34 INT
 35 STO 03
 36 "NO. OF            Prompt and
PIPES?"                store data
 37 PROMPT
 38 STO 06
 39 +
 40 1 E3
 41 /
 42 ST+ 03
 43 ISG 03
 44 1.3
 45 STO 01
 46 "EQUIV.
DIA. ?"
 47 PROMPT
 48 STO 02
 49 "EQUIV.
RUF. ?"
 50 PROMPT
 51 STO 04
                       ------------------
 52♦LBL 03
 53 "L,"               Compute equivalent
 54 XEQ 04             lengths, L and
 55 RCL 02             store L's and
 56 "d,"               Q's
 57 XEQ 04
 58 /
 59 2.63
 60 ENTER↑
 61 .54
 62 /
 63 Y↑X
 64 *
 65 RCL 04
 66 "C,"
 67 XEQ 04
 68 /
 69 .54
 70 1/X
 71 Y↑X
 72 *
 73 STO IND
03
 74 RCL 06
 75 RCL 03
 76 +
 77 "Q,"
 78 XEQ 04
 79 STO IND
Y
 80 ISG 01
 81 ISG 03
 82 GTO 03
                       ------------------
 83 .004
 84 STO 02
 85♦LBL 05
 86 CF IND 0
2
 87 ISG 02             iteration counter
 88 GTO 06             done, display
 89 RCL 00             results
 90 RCL 06
 91 +
 92 1
```

Program Listings

93 +	
94 STO 01	
95 1	
96◆LBL 02	
97 "Q,"	
98 ARCL X	
99 "⊢="	
100 ARCL IND 01	
101 PROMPT	
102 1	
103 +	
104 ISG 01	
105 GTO 02	
106◆LBL 06	
107 SF IND 02	set annunciator
108 RCL 00	
109 6	
110 -	
111 INT	
112 6	
113 +	
114 1 E3	
115 /	
116 7	
117 +	
118 STO 05	
119◆LBL 07	
120 0	
121 STO 03	clear E's
122 STO 04	
123 RCL IND 05	move control reg. into working control reg.
124 STO 01	
125◆LBL 10	
126 XEQ 09	
127 RCL 00	compute ΔQ
128 +	
129 ENTER↑	
130 ENTER↑	
131 RCL 06	
132 +	
133 RCL IND X	
134 ABS	
135 .54	
136 1/X	
137 Y↑X	
138 RCL IND Y	

139 SIGN	
140 *	
141 RCL IND Z	
142 *	
143 ST+ 04	
144 .54	
145 /	
146 RCL IND Y	
147 /	
148 ABS	
149 ST+ 03	
150 GTO 10	
151◆LBL 04	
152 ARCL 01	input prompting
153 "⊢ ?"	
154 PROMPT	
155 RTN	
156◆LBL 09	
157 100	extract pipe numbers
158 ST* 01	
159 RCL 01	
160 INT	
161 ST- 01	
162 X≠0?	done?
163 RTN	no
164 RCL 04	
165 RCL 03	
166 /	
167 FS? 55	print ΔQ so user can watch convergence
168 PRX	
169 INT	
170 STO 03	
171 RCL IND 05	
172 STO 01	adjust Q's by adding ΔQ
173◆LBL 11	
174 100	
175 ST* 01	
176 RCL 01	
177 INT	
178 ST- 01	done?
179 X=0?	yes
180 GTO 08	
181 RCL 00	
182 RCL 06	
183 +	
184 +	
185 RCL 03	
186 ST- IND Y	

Program Listings

```
187 GTO 11
188♦LBL 08
189 ISG 05
190 GTO 07            do next loop
191 GTO 05            iterate
192♦LBL C             compute pressure
193 "EQUIV.           drops
DIA. ?"
194 PROMPT            usually 10
195 STO 02
196 "EQUIV.
RUF. ?"
197 PROMPT            usually 100
198 STO 04
199 1.1
200 STO 01
201♦LBL 12
202 RCL 01            pipe no., n
203 RCL 00
204 +
205 ENTER↑
206 ENTER↑
207 RCL 06
208 +
209 RCL IND           Qn
Y
210 RCL IND           Ln
Y
211 ABS
212 3.5521
213 *
214 RCL 04
215 /
216 .54
217 1/X
218 Y↑X
219 *
220 RCL 02
221 2.63
222 ENTER↑
223 .54
224 /
225 CHS
226 Y↑X
227 *
228 .4335
229 *                 Δ pressure
230 "dP,"
231 FIX 0
232 ARCL 01           display
233 "⊢="              result
234 FIX 2
235 ARCL X
236 PROMPT
237 ISG 01
238 GTO 12
239 .END.
```

60

70

80

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	pointer (last control register No.)		
	working control register		
	Equiv. dia./iteration counter		
	used		
	used		
05	control reg. pointer		
	No. of pipes		
	control reg. 1		
	control reg. 2		
	·		
10	·	60	
	L1		
	L2		
	·		
	·		
15	LN	65	
	Q1		
	Q2		
	·		
	·		
20	QN	70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 7+ TOT. REG. SIZE+67 USER MODE

ENG ___ FIX 2 SCI ___ ON X OFF ___

DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
1	C	1st iteration	
2	C	2nd iteration	
3	C	3rd iteration	
4	C	4th (and last) iteration	
29	C	for proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

PIPE NETWORK ANALYSIS

PROGRAM REGISTERS NEEDED: 67

HEWLETT PACKARD
SOLUTIONS BOOK
FLUID DYNAMICS/HYDRAULICS

ROW 1 (1 : 5)

ROW 2 (5 : 11)

ROW 3 (12 : 19)

ROW 4 (19 : 23)

ROW 5 (24 : 29)

ROW 6 (30 : 33)

ROW 7 (34 : 36)

ROW 8 (36 : 43)

ROW 9 (43 : 46)

ROW 10 (46 : 49)

ROW 11 (49 : 53)

ROW 12 (53 : 58)

ROW 13 (59 : 66)

ROW 14 (66 : 73)

ROW 15 (73 : 80)

ROW 16 (80 : 86)

ROW 17 (87 : 97)

ROW 18 (97 : 104)

ROW 19 (104 : 114)

ROW 20 (114 : 124)

ROW 21 (125 : 134)

ROW 22 (135 : 143)

ROW 23 (143 : 150)

ROW 24 (151 : 157)

ROW 25 (158 : 167)

ROW 26 (168 : 175)

ROW 27 (176 : 186)

ROW 28 (186 : 193)

ROW 29 (193 : 193)

ROW 30 (194 : 196)

ROW 31 (196 : 204)

ROW 32 (205 : 212)

ROW 33 (212 : 221)

ROW 34 (221 : 228)

ROW 35 (228 : 233)

ROW 36 (233 : 239)

ROW 37 (239 : 239)

RESTRICTION METERING ORIFICE CALCULATIONS

This program solves for orifice bore and for differential pressure across an orifice with flange taps for gas flow.

Equations:

$$C = \frac{Q\ NORM}{7727}\sqrt{\frac{T_F\ G}{P_F}}$$

$$Y_P = 1 - [0.333 + 1.145\ (\beta^2 + 0.7\beta^5 + 12\beta^{13})\ \frac{\Delta P}{P_f k}$$

$$S_p = \frac{C}{Y_p\sqrt{h}}\ ,\quad h = 27.7\ \Delta P$$

$$S_p = .58925\beta^2 + 0.2725\beta^3 - 0.825\beta^4 + 1.75\beta^5$$

$$d = \beta D$$

$$S_f = 0.598\beta^2 + 0.01\beta^3 + 0.00001947\beta^2\ (10\beta)^{4.425}$$

$$Y_F = 1 - [\frac{0.41 + 0.35\beta^4}{27.7\ P_f k}]\ h_{normF} \qquad h_{normF} = (C/S_f)^2/Y_F{}^2$$

$$h_{max} = h_{normF}\ [Q_{max}/Q_{normF}]^2$$

Note: If differential pressure across orifice is too high, the second part of the program will not converge.

References: L. K. Spink, Principles and Practices of Flow Meter Engineering, 9th Ed., page 167, Plimpton Press, 1967.
R. G. Cunningham, Orifice Meters with Supercritical Compressible Flow, pages 625-638, ASTM, July 1951.
HP-67/HP-97 Users' Library program #02448D by Larry Richardson.

Example: An orifice is to be sized to control the flow of nitrogen in a 1.939" ID line to 3330 SCFH. A flow transmitter is to be connected to flange taps for an alarm. The temperature is 100°F (560°R), pressure upstream is 8.3 psig (23 psia) pressure downstream is 4.69 psig (P=3.61 psi). Meter maximum is to be 4500 SCFH.

$G = MW_{nitrogen}/MW_{air} \approx 28/28.95 \approx 0.967$

$K_{nitrogen} = 1.4$

Solution:

Keystrokes:	Display:	
[XEQ] [ALPHA] SIZE [ALPHA] 011		
[XEQ] [ALPHA] ORF [ALPHA]	Q ?	
3330 [R/S]	T ?	
560 [R/S]	G ?	
.967 [R/S]	P ?	
23 [R/S]	dP ?	
3.61 [R/S]	K ?	
1.4 [R/S]	D ?	
1.939 [R/S]	d=.585	(" bore)
[R/S]	QMAX ?	
4500 [R/S]	HM=209.	(" water column)

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program			
2	Initialize the program		[XEQ] ORF	Q ?
3	Input: Flow (SCFH);	Q_{norm}	[R/S]	T ?
	Temperature (°R);	T	[R/S]	G ?
	G (air=1);	G	[R/S]	P ?
	upstream pressure (PSIA);	P	[R/S]	dP ?
	pressure difference (PSI);	ΔP	[R/S]	K ?
	K; and	K	[R/S]	D ?
	I.D. of line (inches).	D	[R/S]	d=( ) (inches)
4	Orifice diameter, d, results			
5	To find the transmitter range, nm, input			
	maximum flow	Q_{max}	[R/S]	HM=( ) (inches water column)

Program Listings

Program	Comment	Program	Comment
01♦LBL "ORF"		51 RCL 10	
02 "Q ?"		52 X↑2	
03 PROMPT		53 +	
04 "T ?"		54 1.145	
05 PROMPT	Prompt and store	55 *	
06 "G ?"	data	56 .333	
07 PROMPT		57 +	
08 "P ?"		58 RCL 01	
09 PROMPT		59 *	
10 STO 00		60 RCL 00	
11 /		61 /	
12 *		62 -	Y_p
13 SQRT		63 RCL 09	
14 X<>Y		64 X<>Y	
15 STO 03		65 /	
16 *		66 STO 04	Calculate S_p
17 STO 09		67 .5	
18 "dP ?"		68 STO 05	β
19 PROMPT		69 STO 06	
20 FIX 3		70♦LBL 01	
21 STO 01		71 RCL 05	
22 27.7		72 RCL 05	
23 *		73 RCL 05	
24 SQRT		74 1.75	
25 ST/ 09		75 *	
26 "K ?"		76 .825	
27 PROMPT		77 -	
28 ST* 00		78 *	
29 "D ?"		79 .2725	
30 PROMPT		80 +	
31 STO 02		81 *	
32 X↑2		82 .58925	
33 7727		83 +	
34 *		84 *	
35 ST/ 09		85 *	S_p=power series
36 .5	guess for	86 RCL 04	
37 STO 10	β	87 -	
38♦LBL 00		88 X=0?	
39 1		89 GTO 03	
40 RCL 10	Calculate d	90 X<0?	iterate
41 13		91 GTO 04	
42 Y↑X		92 RCL 06	
43 12		93 ST- 05	
44 *		94 GTO 05	
45 RCL 10		95♦LBL 04	
46 5		96 RCL 06	
47 Y↑X		97 ST+ 05	
48 .7		98♦LBL 05	
49 *		99 RCL 06	
50 +		100 RCL 02	
		101 *	

Program Listings

RND	
X=0?	
GTO 03	
2	
ST/ 06	
GTO 01	
◆LBL 03	
RCL 10	
RCL 05	
-	
RCL 02	
*	
RND	
X=0?	
GTO 06	
RCL 05	
STO 10	
GTO 00	
◆LBL 06	
RCL 05	β
RCL 02	D
*	----------------
"d="	
ARCL X	Display
PROMPT	result
FIX 0	----------------
"QMAX ?"	Calculate h_m
PROMPT	
STO 04	
RCL 01	
27.7	
*	
SQRT	
RCL 09	
*	
STO 10	C
RCL 05	β
4	
Y↑X	
.35	
*	
.41	
+	
RCL 00	
/	
27.7	
/	
STO 01	[ ] Y_F
RCL 05	
STO 02	
RCL 02	β

X↑2	
.598	
*	
RCL 02	
3	
Y↑X	
.01	
*	
+	
RCL 02	
6.425	
Y↑X	
.51804	
*	
+	
RCL 10	S_f
X<>Y	
/	
X↑2	
STO 10	
STO 07	
◆LBL 07	
RCL 10	
1	
RCL 01	
RCL 07	
*	
-	
X↑2	
/	
STO 08	h_{normF}
RCL 07	
-	
RND	
X=0?	
GTO 08	
RCL 08	
STO 07	
GTO 07	
◆LBL 08	
RCL 04	Q_{max}
RCL 03	
/	
X↑2	
RCL 08	
*	----------------
"HM="	
ARCL X	Display
PROMPT	result
.END.	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	KP	50	
	used		
	used		
	QMAX		
	used		
05	Bnew	55	
	Bincrement		
	old h norm		
	new h norm		
	used		
10	used	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 011 TOT. REG. 056 USER MODE

ENG ____ FIX 3 SCI ____ ON ____ OFF x

DEG x RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

RESTRICTION METERING
ORIFICE CALCULATIONS
PROGRAM REGISTERS NEEDED: 46

ROW 1 (1 – 4)

ROW 2 (4 – 8)

ROW 3 (9 – 18)

ROW 4 (18 – 25)

ROW 5 (26 – 31)

ROW 6 (32 – 39)

ROW 7 (40 – 49)

ROW 8 (50 – 56)

ROW 9 (56 – 67)

ROW 10 (68 – 76)

ROW 11 (76 – 82)

ROW 12 (82 – 90)

ROW 13 (91 – 99)

ROW 14 (100 – 109)

ROW 15 (110 – 120)

ROW 16 (121 – 128)

ROW 17 (128 – 133)

ROW 18 (134 – 143)

RESTRICTION METERING ORIFICE CALCULATIONS

ROW 19 (143 - 152)

ROW 20 (153 - 160)

ROW 21 (161 - 165)

ROW 22 (165 - 177)

ROW 23 (178 - 189)

ROW 24 (190 - 199)

ROW 25 (199 - 202)

ENERGY EQUATION FOR STEADY FLOW

Given any eight of the nine terms in the equation below, the program calculates the unknown term.

Energy Equation:

$$\frac{P_1}{W} + Z_1 + \frac{V_1^2}{2g} + H_p = \frac{P_2}{W} + Z_2 + \frac{V_2^2}{2g} + H_T + H_L$$

where

H = total dynamic head, ft.

H_p = head added by pump, ft.

H_T = head extracted by turbine, ft.

H_L = head loss due to friction, ft.

$\frac{V^2}{2g}$ = velocity head, ft.

$\frac{P}{W}$ = pressure head, ft.

Z = potential head, ft.



References: HP-67/HP-97 Users' Library program #00267D
Fluid Mechanics and Hydraulics, by Ronald V. Giles, Schaums Outline Series, McGraw-Hill Book Company, New York, 1962.

Example:



Find the head extracted by the turbine.

Solution:

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 013	
[XEQ] [ALPHA] ENRG [ALPHA]	P1 HEAD ?
0 [R/S]	Z1 HEAD ?
400 [R/S]	V1 HEAD ?
0 [R/S]	HP HEAD ?
0 [R/S]	P2 HEAD ?
0 [R/S]	Z2 HEAD ?
10 [R/S]	V2 HEAD ?
100 [ENT] 12.48 [÷] [x^2] 64.4 [÷] [R/S]	HT HEAD ?
[R/S]	HL HEAD ?
25 [R/S]	HT HEAD=364.00

User Instructions

SIZE: 013

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program			
2	Initialize the program		[XEQ] ENRG	P1 HEAD ?
3	Input 8 of the following:	P1 HEAD	[R/S]	Z1 HEAD ?
		Z1 HEAD	[R/S]	V1 HEAD ?
		V1 HEAD	[R/S]	HP HEAD ?
		HP HEAD	[R/S]	P2 HEAD ?
		P2 HEAD	[R/S]	Z2 HEAD ?
		Z2 HEAD	[R/S]	V2 HEAD ?
		V2 HEAD	[R/S]	HT HEAD ?
		HT HEAD	[R/S]	HL HEAD ?
		HL HEAD	[R/S]	(xx) HEAD=( )
	When prompted for the unknown variable,			
	press [R/S] (make no input). That			
	quantity is then automatically calculated.			

Program Listings

Initialize

```
 01♦LBL "ENR
G"
 02 CF 02
 03 SF 01
 04 1.1
 05 STO 00
 06 CF 22
```

Prompt and store data

```
 07 "P1"
 08 XEQ 15
 09 "Z1"
 10 XEQ 15
 11 "V1"
 12 XEQ 15
 13 "HP"
 14 XEQ 15
 15 "P2"
 16 XEQ 15
 17 "Z2"
 18 XEQ 15
 19 "V2"
 20 XEQ 15
 21 "HT"
 22 XEQ 15
 23 "HL"
 24 XEQ 15
```

Compute unknown

```
 25 XEQ 16
 26 RCL 11
 27 STO 12
 28 XEQ 18
 29 RCL 12
 30 FC? 02
 31 X<>Y
 32 -
```

Display result

```
 33 CLA
 34 ARCL 10
 35 "⊢ HEAD=
"
 36 ARCL X
 37 PROMPT
```

Input storage

```
 38♦LBL 15
 39 FS? 01
 40 ASTO 10
 41 "⊢ HEAD
?"
 42 PROMPT
 43 STO IND
00
```

Zero results

```
 44 FS?C 22
 45 GTO 10
 46 ST- IND
00
```

RHS or LHS?

```
 47 CF 01
 48 RCL 00
 49 5
 50 X<=Y?
 51 SF 02
 52♦LBL 10
 53 ISG 00
 54 RTN
```

Calculate RHS

```
 55♦LBL 16
 56 5.009
 57 STO 00
 58 0
 59♦LBL 17
 60 RCL IND
00
 61 +
 62 ISG 00
 63 GTO 17
 64 STO 11
 65 RTN
```

Calculate LHS

```
 66♦LBL 18
 67 1.004
 68 STO 00
 69 0
 70 GTO 17
 71 .END.
```

80

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	pointer	50	
	P1 head		
	Z1 head		
	v1 head		
	HP head		
05	P2 head	55	
	Z2 head		
	V2 head		
	HT head		
	HL head		
10	prompt	60	
	temp. storage		
	temp. storage		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 013 TOT. REG. 038 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF x

DEG x RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01	S	store prompt	don't store prompt
02	C	unknown on LHS	unknown on RHS
22	C	input made	no input made

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ENERGY EQUATION FOR
STEADY FLOW
PROGRAM REGISTERS NEEDED: 25

ROW 1 (1 - 4)

ROW 2 (4 - 9)

ROW 3 (9 - 13)

ROW 4 (14 - 18)

ROW 5 (18 - 22)

ROW 6 (22 - 28)

ROW 7 (28 - 35)

ROW 8 (35 - 40)

ROW 9 (40 - 43)

ROW 10 (44 - 51)

ROW 11 (52 - 58)

ROW 12 (59 - 66)

ROW 13 (66 - 71)

ROW 14 (71 - 71)

COMPRESSIBLE FLOW IN VARIABLE AREA DUCTS

This program solves the area ratio mach number relationship for isentropic flow of a perfect gas in a variable area duct. The program will find M given A/A* or A/A* given M, or T/To, P/Po or M given any one of these three quantities. The zero subscript refers to stagnation conditions.

Equations:

$$A/A^* = \frac{1}{M}\left(\frac{1 + \frac{k-1}{2} M^2}{\frac{k+1}{2}}\right)^{\frac{k+1}{2(h-1)}}$$

$$P/Po = \left(1 + \frac{k-1}{2} M^2\right)^{\frac{-k}{h-1}}$$

$$T/To = \left(1 + \frac{k-1}{2} M^2\right)^{-1}$$

where

T = temperature

P = pressure

A = cross-sectional area of ducts at which M occurs

M = Mach number

A* = throat area of duct

and

$k = \frac{Cp}{Cn}$, Cp = specific heat at constant pressure
Cn = specific heat at constant volume

Notes: The equations apply only to a perfect gas with constant specific heats. An initial guess for M must be supplied for the first equation given above if the area ratio A/A* is the known quantity. If the guess for M is < 1 then the program converges to a Mach number < 1. If the guess for M is > 1 then the program converges to a solution for M > 1.

References: HP-65 Users' Library program #00780A by Harry Townes.

Example:



For K = 1.4 and A/A* = 1.5, find the two possible Mach numbers, T/To and P.Po.

Solution:

Keystrokes:	Display:	
[USER]		(Set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 009		
[XEQ] [ALPHA] COMFLO [ALPHA]	LBL A OR B ?	
[A]	K ?	
1.4 [R/S]	A/A* ?	
1.5 [R/S]	GUESS FOR M ?	
.5 [R/S]	M=0.4303	
[A]	K ?	
1.4 [R/S]	A/A* ?	
1.5 [R/S]	GUESS FOR M ?	
1.5 [R/S]	M=1.8541	
[B]	K ?	
1.4 [R/S]	M ?	
.4303 [R/S]	T/To=0.9643	
[R/S]	P/Po=0.8805	
[B]	K ?	
1.4 [R/S]	M ?	
1.8541 [R/S]	T/To=0.5926	
[R/S]	P/Po=0.1602	

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program and set USER mode		[USER]	
2	Initialize the program		[XEQ] COMFLO	LBL A OR B ?
3	To find A/A* or M given the other			
	quantity, press →		[A]	K ?
3a	Input K	K	[R/S]	A/A* ?
3b	Input A/A* (note 1)	A/A*	[R/S]	M ?
				-or- GUESS FOR M ?
3c	Input M or a guess for M	M	[R/S]	A/A*=( )
	If the guess for M is < 1 then the result-			-or- M=( )
	ing M is < 1. If the guess is > 1 then			
	the result is > 1. The equation has two			
	roots when A/A* is the known value.			
4	To find T/To, P/Po, or M given any one			
	of these three quantities, press →		[B]	K ?
4a	Input K	K	[R/S]	M ?
4b	Input M and find T/To and P/Po (note 1)	M	[R/S]	T/To=( )
			[R/S]	P/Po=( )
4c	Input T/To and find M (note 1)	T/To	[R/S]	M=( )
4d	Input P/Po and find M	P/Po	[R/S]	M=( )
	Note 1: When prompted for an unknown			
	quantity, press [R/S] (make no			
	input).			

Program Listings

Program	Comment
01♦LBL "COM FLO"	
02 "LBL A OR B ?"	
03 PROMPT	
04♦LBL A	A/A* vs. M
05 CF 01	
06 "K ?"	
07 PROMPT	
08 1	
09 +	
10 2	
11 /	
12 STO 03	
13 RCL 03	
14 1	
15 -	
16 STO 04	
17 /	
18 2	
19 /	
20 STO 02	
21 CF 22	
22 "A/A* ?"	
23 PROMPT	
24 FS? 22	Input made?
25 SF 01	yes
26 STO 05	
27 "GUESS FOR M?"	
28 FC? 22	
29 "M ?"	
30 PROMPT	
31 STO 01	
32♦LBL 01	
33 RCL 01	Calculate A/A*
34 X↑2	
35 RCL 04	
36 *	
37 1	
38 +	
39 RCL 03	
40 /	
41 STO 06	
42 RCL 02	
43 Y↑X	
44 STO 07	
45 RCL 01	
46 /	A/A*
47 FC? 01	
48 GTO 02	

Program	Comment
49 RCL 05	
50 -	Calculate M
51 STO 08	iteratively
52 RCL 07	
53 CHS	
54 RCL 01	
55 X↑2	
56 /	
57 RCL 06	
58 RCL 02	
59 1	
60 -	
61 Y↑X	
62 +	
63 RCL 08	
64 X<>Y	
65 /	
66 ST- 01	
67 ABS	
68 RCL 01	
69 /	
70 1 E-3	
71 X<=Y?	
72 GTO 01	
73 RCL 01	
74 GTO 09	
75♦LBL 02	
76 "A/A*"	
77 GTO 03	
78♦LBL B	T/To, P/Po, vs. M
79 "K ?"	
80 PROMPT	
81 STO 00	
82 1	
83 -	
84 ST/ 00	
85 2	
86 /	
87 STO 02	
88 CF 22	
89 "M ?"	
90 PROMPT	
91 FS? 22	M input?
92 GTO 04	yes
93 "T/Ta ?"	
94 PROMPT	
95 FS? 22	
96 GTO 05	
97 "P/Pa ?"	
98 PROMPT	
99 RCL 00	

Program Listings

Steps	Comments
100 CHS 101 1/X 102 Y↑X 103 1 104 - 105 RCL 02 106 / 107 SQRT 108♦LBL 09 109 "M"	Calculate M given P/Po
110♦LBL 03 111 "⊢=" 112 ARCL X 113 PROMPT 114 RTN	Display routine
115♦LBL 04 116 X↑2 117 * 118 1 119 + 120 STO 01 121 1/X 122 "T/Ta" 123 XEQ 03	Calculate T/To
124 RCL 01 125 RCL 00 126 Y↑X 127 1/X 128 "P/Pa" 129 XEQ 03	Calculate P/Po
130♦LBL 05 131 1/X 132 1 133 - 134 RCL 02 135 / 136 SQRT 137 GTO 09 138 .END.	Calculate M given T/To

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	K/(K-1)	50	
	Used		
	(K+1)/2(K-1) or (K-1)/2		
	(K_1)/2		
	(K-L)/2		
05	A/A*	55	
	Used		
	Used		
	Used		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 052 USER MODE

ENG ___ FIX 4 SCI ___ ON x OFF ___

DEG x RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01	C	calculate M	calculate A/A*
22	C	refer to owner's manual	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

COMPRESSIBLE FLOW IN VARIABLE
AREA DUCTS
PROGRAM REGISTERS NEEDED: 36

ROW 1 (1 – 2)

ROW 2 (2 – 4)

ROW 3 (5 – 13)

ROW 4 (14 – 22)

ROW 5 (22 – 27)

ROW 6 (27 – 29)

ROW 7 (29 – 40)

ROW 8 (41 – 51)

ROW 9 (52 – 64)

ROW 10 (65 – 72)

ROW 11 (73 – 78)

ROW 12 (79 – 87)

ROW 13 (88 – 93)

ROW 14 (93 – 97)

ROW 15 (97 – 106)

ROW 16 (107 – 115)

ROW 17 (116 – 123)

ROW 18 (123 – 129)

COMPRESSIBLE FLOW IN VARIABLE
AREA DUCTS

ROW 19 (130 – 138)

FLOOD ROUTING AND HYDROGRAPHS

This program calculates either a unit hydrograph or a soil conservation service hydrograph from a given peak time (time of concentration) and peak flow. Any time interval can be selected. The program will also route a given hydrograph through a given dam calculating an outflow hydrograph from given storage conditions and a given outflow structure.

Equations:

Flood routing

$$I_n + I_{n+1} + \frac{2\ S_n}{\Delta t} - 0 = \frac{2\ S_{n+1}}{\Delta t} + 0_{n+1}$$

where:

I = inflow;

S = storage;

Δt = time interval;

0 = outflow;

n = cycle number.

UNIT HYDROGRAPH

$y = 1.45x^{1.67}$, $0.5 > x > 0$

$y = 1.16 + \ln x$, $0.9 > x > 0.5$

$y = \text{Sin}\ (e^{x-1} \cdot 90)$, $1.2 > x > 0.9$

$y = 1.93 - .83\ x$, $1.6 > x > 1.2$

$y = 7.49e^{-1.63x}$, $x > 1.6$

SCS HYDROGRAPH

$y = 1.7\ x^2,\ 0.7 > x > 0$

$y = 1.06 + .8 \ln x,\ 1. > x > 0.7$

$y = 1.9 - .83x,\ 1.8 > x > 1.0$

$y = 5.7e^{-1.44x},\ x > 1.8$

where x = time/time of peak

and y = flow/peak flow

Notes: If a subroutine is used to calculate outflow from [$\frac{2\ S}{\Delta t}$ + outflow vs. outflow] it must always produce flows greater than or equal to zero.

References:

1. HP-67/HP-97 USER'S LIBRARY program #01442D by Lawrence Busack

2. PENNSYLVANIA STATE UNIVERSITY, Hydrologic and Hydraulic Analysis for Small Watersheds, PENN State University, University Park, PA, 1974.

3. U.S. Department of Interior, Bureau of Reclamation, Design of Small Dams, GPO, Washington, D.C., 1974.

4. U.S. Department of Agriculture, Soil Conservation Service, National Engineering Handbook Section 4 - Hydrology, GPO, Washington, D.C., 1972.

Example:



A dam with a sharp crested weir spillway is located on a small stream. The spillway is 100 feet wide and has a discharge coefficient of 3.5. At the elevation of the spillway no water is stored. At an elevation of 2 feet above the spillway 100 acre-feet of water is stored and at an elevation of 4 feet above the spillway 250 acre-feet of water is stored. A flood which reaches a peak flow of 1400 cfs in 2.9 hours flows through the dam. Calculate the inflow hydrograph using the SCS curve and the outflow hydrograph. Use a 1/4 hour time interval.

Solution:

First develop the outflow vs. $\frac{2S}{\Delta t}$ + outflow relationship

at 2 feet elevation,

$$\text{outflow} = (3.5)\ (100)\ (2)^{3/2}\quad [Q=CLH^{3/2}]$$

$$= 989.95 \text{ cfs}$$

$$\frac{2S}{\Delta t} + \text{outflow} = \left[\frac{(2)\ (100)\ \text{AC-FT})}{.25 \text{ HRS}}\right] \left[43560 \frac{\text{ft}^3}{\text{AC-FT}}\right]$$

$$\text{X} \left[\frac{1}{3600 \frac{\text{SEC}}{\text{HR}}}\right] + 989.95$$

$$= 10669.95 \text{ cfs}$$

Then, calculating outflow values at each elevation and $\frac{2S}{\Delta t}$ + outflow (as above) for 4 feet and interpolating to find intermediate values, the following table is developed:

ELEVATION	OUTFLOW	$\frac{2S}{\Delta t}$ + OUTFLOW
0	0	0
1/2	123.74	2543.74
1	350.00	5190.00
1 1/2	642.99	7902.99
2	989.95	10669.95
2 1/2	1383.50	14693.50
3	1818.65	18758.65
3 1/2	2291.77	22861.77
4	2800.00	27000.00

Using a curve-fitting program to yield an analytical expression for the above table,

$$\text{OUTFLOW} \approx 0.11 \left[\frac{2S}{\Delta t} + \text{outflow} \right] - 200$$

This equation is programmed in subroutine 09.

The hydrographs (inflow and outflow) are then calculated. The keystrokes which follow reflect a printer in the system.

KEYSTROKES:	DISPLAY:	
[USER]		(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 009		
[XEQ] [ALPHA] HYDRO [ALPHA]	TIME INTVL. ?	
.25 [R/S]	TIME, PEAK Q ?	
2.90 [R/S]	PEAK Q ?	
1400 [R/S]	A,B, OR C ?	
[C]	A OR B ?	
[B]	T=0.00	
	IN=0.00	
	T=0.25	
	IN=17.69	

```
OUT=0.00
•
•          (increment T)
•
T=1.25

IN=442.18

OUT=0.00

T=1.50

IN=636.74

OUT=86.00

etc.
•
•          (increment T)
•
```

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program and set USER mode		[USER]	
2	Initialize the program		[XEQ]HYDRO	TIME INTVL. ?
3	Input: time interval;	i	[R/S]	TIME, PEAK Q ?
	time of peak flow;	t peak	[R/S]	PEAK Q ?
	and peak flow.	Q peak	[R/S]	A, B, OR C ?
4	To do flood routing, press		[C]	A OR B ?
5	For a unit hydrograph, press		[A]	T=( )
6	find inflow		[R/S]*	IN=( )
7	flood routing only, find outflow		[R/S]*	OUT=( )
8	Increment T and go to step 6		[R/S]*	T=( )
9	For an SCS hydrograph, press		[B]	T=( )
10	Find inflow		[R/S]*	IN=( )
11	flood routing only, find outflow		[R/S]*	OUT=( )
12	Increment T and go to step 10		[R/S]*	T=( )
	NOTE: If an analytical expression is not			
	known for outflow as a function of $\frac{2S}{\Delta t}$ +			
	outflow, outflow may be read from a graph			
	of the function and input manually at step			
	7 or 11. To do so, change line 112 from			
	XEQ09 to R/S. When the program stops,			
	enter a value for outflow which corresponds			
	to the value for $\frac{2S}{\Delta t}$ + outflow in the display,			
	and press [R/S] to continue.			
	*These keystrokes are unnecessary if there			
	is a printer in the system.			

Program Listings

Initialize

```
01♦LBL "HYDRO"
02 CLRG
03 SF 21
04 CF 02
```

Prompt and store data

```
05 "TIME INTVL. ?"
06 PROMPT
07 STO 01
08 "TIME, PEAK Q?"
09 PROMPT
10 STO 00
11 "PEAK Q ?"
12 PROMPT
13 STO 04
```

UNIT hydrograph / SCS hydrograph / do flood routing

```
14 "A,B, OR C ?"
15 PROMPT
16♦LBL A
17 11
18 STO 08
19 GTO 11
20♦LBL B
21 12
22 STO 08
23 GTO 12
24♦LBL C
25 SF 02
26 SF 03
27 "A OR B ?"
28 PROMPT
```

Display routine

```
29♦LBL 10
30 "├="
31 ARCL X
32 AVIEW
33 RTN
```

SCS hydrograph calculations

```
34♦LBL 12
35 RCL 02
36 RCL 00
37 /
38 STO 03
39 .7
40 X>Y?
41 GTO 00
42 X<>Y
43 1
44 X>Y?
45 GTO 01
46 X<>Y
47 1.8
48 X>Y?
49 GTO 02
50 RCL 03
51 1.44
52 *
53 CHS
54 E↑X
55 5.7
56 *
57 GTO 03
58♦LBL 00
59 RCL 03
60 X↑2
61 1.7
62 *
63 GTO 03
64♦LBL 01
65 RCL 03
66 LN
67 .8
68 *
69 1.06
70 +
71 1
72 X>Y?
73 X<>Y
74 GTO 03
75♦LBL 02
76 RCL 03
77 .83
78 *
79 CHS
80 1.9
81 +
82 1
83 X>Y?
84 X<>Y
```

increment time

```
85♦LBL 03
86 RCL 04
87 *
88 RCL 02
89 "T"
90 XEQ 10
91 RCL 01
92 +
93 STO 02
94 X<>Y
95 "IN"
96 XEQ 10
```

Program Listings

```
 97 FC? 02
 98 GTO IND
08
 99 FS?C 03
100 GTO 13
101 RCL 05
102 X<>Y
103 STO 05
104 +
105 STO 06
106 RCL 07
107 FS?C 01
108 RCL 05
109 RCL 06
110 +
111 STO 07
112 XEQ 09
```

change to STOP for manual input

```
113 2
114 *
115 ST- 07
116 2
117 /
118 "OUT"
119 XEQ 10
120 GTO IND
08
121+LBL 13
122 SF 01
123 STO 05
124 GTO IND
08
125+LBL 11
```

UNIT hydrograph calculations

```
126 RCL 02
127 RCL 00
128 /
129 STO 03
130 .5
131 X>Y?
132 GTO 04
133 X<>Y
134 .9
135 X>Y?
136 GTO 05
137 X<>Y
138 1.2
139 X>Y?
140 GTO 06
141 X<>Y
142 1.6
143 X>Y?
144 GTO 07
145 RCL 03
146 1.63
147 *
148 CHS
149 E↑X
150 7.49
151 *
152 GTO 03
153+LBL 04
154 RCL 03
155 1.67
156 Y↑X
157 1.45
158 *
159 GTO 03
160+LBL 05
161 RCL 03
162 LN
163 1.16
164 +
165 GTO 03
166+LBL 06
167 RCL 03
168 1
169 -
170 E↑X
171 90
172 *
173 DEG
174 SIN
175 GTO 03
176+LBL 07
177 RCL 03
178 .83
179 *
180 CHS
181 1.93
182 +
183 GTO 03
184+LBL 09
185 .11
186 *
187 200
188 -
189 X<0?
190 0
191 RTN
192 .END.
```

calculate outflow peculiar to the sample problem

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

#		#	
00	Time, peak	50	
	Time increment		
	Time, total		
	T/T peak		
	Q peak		
05	Qi	55	
	In+In+1		
	2S/Δt + 0,-0		
	Pointer		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 59 USER MODE
ENG ____ FIX 2 SCI ____ ON X OFF ____
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01	C	1st pass through	
02	C	do flood routing	
03	C	1st pass through	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FLOOD ROUTING AND HYDROGRAPHS

PROGRAM REGISTERS NEEDED: 51

ROW 1 (1 – 4)

ROW 2 (4 – 5)

ROW 3 (5 – 8)

ROW 4 (8 – 11)

ROW 5 (11 – 14)

ROW 6 (14 – 20)

ROW 7 (20 – 27)

ROW 8 (27 – 30)

ROW 9 (31 – 41)

ROW 10 (41 – 49)

ROW 11 (50 – 57)

ROW 12 (57 – 66)

ROW 13 (67 – 74)

ROW 14 (75 – 83)

ROW 15 (84 – 93)

ROW 16 (94 – 99)

ROW 17 (100 – 110)

ROW 18 (111 – 118)

ROW 19 (118 – 125)

ROW 20 (126 – 135)

ROW 21 (136 – 142)

ROW 22 (143 – 150)

ROW 23 (150 – 157)

ROW 24 (157 – 163)

ROW 25 (164 – 174)

ROW 26 (175 – 181)

ROW 27 (182 – 189)

ROW 28 (190 – 192)

NOTES

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41 programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41, HP has an extensive library of "Application Pacs". These programs transform your HP-41 into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, miniature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41.

You can choose from:

Aviation (Pre-Flight Only) 00041-15018
Clinical Lab 00041-15024
Circuit Analysis 00041-15024
Financial Decisions 00041-15004
Mathematics 00041-15003
Structural Analysis 00041-15021
Surveying 00041-15005
Securities 00041-15026
Statistics 00041-15002
Stress Analysis 00041-15027
Games 00041-15022
Home Management 00041-15023
Machine Design 00041-15020
Navigation 00041-15017
Real Estate 00041-15016
Thermal and Transport Science 00041-15019
Petroleum Fluids 00041-15039

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41 Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales 00041-90094
Home Construction Estimating 00041-90096
Lending, Saving and Leasing 00041-90086
Real Estate 00041-90136
Small Business 00041-90137
Geometry 00041-90084
High-Level Math 00041-90083
Test Statistics 00041-90082
Antennas 00041-90093
Chemical Engineering 00041-90100
Control Systems 00041-90092
Electrical Engineering 00041-90088
Fluid Dynamics and Hydraulics 00041-90139
Games II 00041-90443
Civil Engineering 00041-90089
Heating, Ventilating & Air Conditioning 00041-90140
Mechanical Engineering 00041-90090
Solar Engineering 00041-90138
Calendars 00041-90145
Cardiac/Pulmonary 00041-90097
Chemistry 00041-90102
Games 00041-90099
Optometry I (General) 00041-90143
Optometry II (Contact Lens) 00041-90144
Physics 00041-90142
Surveying 00041-90141
Time Module Solutions 00041-90395

*Some books require additional memory modules to accomodate all programs.

FLUID DYNAMICS AND HYDRAULICS

CONDUIT FLOW
FLOW WITH A FREE SURFACE
PIPE SLIDE-RULE
FORCES AT BENDS AND FITTINGS
VALVE SIZING
PIPE NETWORK ANALYSIS
RESTRICTION METERING ORIFICE CALCULATIONS
ENERGY EQUATION FOR STEADY FLOW
COMPRESSIBLE FLOW IN VARIABLE AREA DUCTS
FLOOD ROUTING AND HYDROGRAPHS

HEWLETT PACKARD

00041-90139 Rev. F 1/84 Printed in U.S.A.